
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc74681324]Corrections required
0.1 [bookmark: _Toc74681325]Correction to function fl_TC_11_3_9_TestBody
	Function name
	fl_TC_11_3_9_TestBody

	Reason for change
	1. According to CR R5-213507, the time should be 11.2s.
2. According to TS 24.501 cl 4.5.4.2, the barring is alleviated for the access category 33 when UE has camped on NR Cell 12 at step 6, UE in RRC_CONNECTED state may initiate the NAS transport procedure to send the PDU SESSION ESTABLISHMENT REQUEST message. Optional behaviour for UE shall be supplemented to resolve this issue.
Note :it is supported with a prose CR.

	Summary of change
	1. Modify the time 20 to 11.2.
2. Modify the procedure of step6.

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	  function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (omit) DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;
    timer t_Wait := 20.0;
    
    v_DNN := cs_DNN(-, f_DomainName_Encode(tsc_DNNValue)); // @sic R5s210307 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    
   //@siclog "Step 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               // @sic R5s210307 sic@
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
     // @siclog "Step 6" siclog@
    //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection.
    f_NR5GC_MobilityRegistration(nr_Cell12);
    
    //@siclog "Step 7" siclog@
    //Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data within 20s?
    t_Wait.start;
    v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
    if (v_ReceiveRRCSetupRequest) {
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    } else {
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }

    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 R5s210307 sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);
    
  }



After Change:
	  function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (omit) DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;
    timer t_Wait; 
    
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;    
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    
    
    v_DNN := cs_DNN(-, f_DomainName_Encode(tsc_DNNValue)); // @sic R5s210307 sic@
    
    //@siclog "Step 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
    
   //@siclog "Step 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start(11.2);
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               // @sic R5s210307 sic@
               f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
     // @siclog "Step 6" siclog@
    //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection.
//REMOVE  f_NR5GC_MobilityRegistration(nr_Cell12,-, -, -, -, noRrcConnectionRelease); 
   
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell12, tsc_NR_RRC_TI_Def, ?);
	
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
	
    f_NR_RRC_ActivateSecurity (nr_Cell12, valueof(v_ReceivedMsg.SecurityProtection.NasCount)); 
    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease);

	v_SRB:= tsc_NR_RbId_SRB1;
	//EXCEPTION: Steps 6Aa1-6Ab describes optional behaviour that depends on the UE implementation.
    t_Wait.start(10.0);
	
    alt{
		//Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN value in UL NAS TRANSPORT message is the same as the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell12, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
              t_Wait.stop;
			  f_NR5GC_PDUSessionEstablishment_InnerLoop(nr_Cell12, v_ReceivedAsp, SRB1, omit, -, v_DNN, -, v_DNN, -, -, -, -, -, -, -, -, -, -, -, -, -, -, -, cr_NG_Request_Type('001'B));
			}
        []t_Wait.timeout
			{
			   f_NR_RRCRelease(nr_Cell12);
			   
			        //@siclog "Step 6Ab3" siclog@
				   //Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
				   f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, tsc_DNNValue);
				   
			        //@siclog "Step 6Ab4" siclog@
				   //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data within 20s?
				   t_Wait.start(20.0);
				   v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
				   if (v_ReceiveRRCSetupRequest) {
						f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
				   } else {
						f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
				   }
			   
			    	//@siclog "Step 6Ab5-6Ab10" siclog@
				   //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
				   f_NR_RRCSetup_Def (nr_Cell12);
				   
				   v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
														 cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
														 tsc_SHT_IntegrityProtected);
														 
				   if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 R5s210307 sic@
						f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
				   }
				   
				   f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);
				   
				   v_SRB:= tsc_NR_RbId_SRB2;				   
    //@siclog "Steps 15 - 16" siclog@
                  //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
                  //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
                  f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, v_SRB, -, - , v_DNN, -, v_DNN);
		    }
    }
   
  }



